COMMONWEALTH OF AUSTRALIA

DEPARTMENT OF NATIONAL DEVELOPMENT

BUREAU OF MINERAL RESOURCES, GEOLOGY AND GEOPHYSICS

Record No. 1971/74

A Fortran Program for Calculating the Gravity Effect of Vertical Cylinders

by

J. P. Cull

RECORD NO. 1971/74

A FORTRAN PROGRAM FOR CALCULATING THE GRAVITY EFFECT OF VERTICAL CYLINDERS

by

J. P. CULL

CONTENTS

ILLUSTRATIONS

SUMMARY

A description is given of a computer program used for simulating gravity anomalies over vertical cylindrical bodies.

All data are input on cards to a CDC 3600 computer and the output is a plot of the gravity profile (using a CALCOMP 565 plotter, 12 in) and a print-out of the profile values.

If plotting facilities are not available output may be confined to a print-out by the removal of the plotting subroutine.

The program may be used to compute gravity profiles over complex circularly symmetrical bodies, such as Gosses Bluff.

1. INTRODUCTION

The computation of the gravity effects of two dimensional bodies by machine methods is well established (Morgan & Grant, 1963; Talwani, Worzel & Landisman, 1959; Haigh, Pollard & Williams, in prep.) and several computer programs for this purpose are in operation. In many cases structure can not be approximated by two-dimensional (2-D) cross-sections and gravity computations must be extended to three dimensions. However, existing programs for three-dimensional (3-D) bodies require lengthy execution time, and data defining the body limits are complex. There are many 3-D geological structures which may be approximated by regular geometrical figures. Such regular bodies are easily specified and their gravity effects are conveniently found by machine summation of line-mass effects. The summation process eliminates complex approximation formulae used in existing 3-D programs.

The program GRAVCYL was developed to compute gravity profiles across finite vertical cylinders. Specification data are simple and execution time is short. The program is useful for geological structures such as ring-dykes, volcanic plugs, and astroblemes which show circular symmetry in plan and which may be approximated by summing and subtracting vertical cylinders (or discs) of various dimensions and densities. The gravity profiles over such structures may then be obtained by summing the effects of all individual cylinders. The program may also be used to calculate the gravity effects at the edges of bodies, which have vertical sides and boundaries that can be approximated by arcs of circles.

The cylinders must be concentric but may be displaced along the bisector of the traverse line.

2. MECHANISM OF COMPUTATION

To compute the vertical component of gravity at any point on the surface, the cylinder is first divided into a large number of vertical square rods, which can then be considered as line masses. The effect of each line mass is then computed according to the formula (derived in Appendix 1):

$$g_v = Gm_1 \ (1/R_1 - 1/R_2)$$

where R_1 and R_2 are the distances from the observing point to the top and bottom of the rod respectively; m_1 is the mass per unit length of the rod. Then the total gravity effect of the cylinder (in a vertical direction) at that observation point is just the summation of the individual g_v calculated for each line mass.

The observation point is then moved along the traverse by an amount DX; the observation number NL is incremented, and the previous calculations are repeated. As the gravity effect at each point is calculated it is stored in the matrices GO (KX, NL) and VG (NL). At the end of the traverse a new card is read to obtain the specifications of the next cylinder in the composite model, and the KX (cylinder number) is incremented. The gravity effects of this new cylinder are calculated at the previous observation points and are again stored in GO and VG after incrementing KX. Hence, the 2-D array GO contains traverse results for each individual cylinder, and the matrix VG contains the resultant composite profile.

When an EOF card is read, the subroutines stating PLOTOUT and PRINTOUT are called up, and the output of the profiles is commenced.

Because square rods are used to build up the cylinders, errors occur at the edges. In this program, the rods were allowed to extend beyond the circumference so that the line masses were centred on the boundary; this procedure leads to a small mass excess.

The accuracy of the result is governed by the number of vertical square rods comprising the cylinder. The radius of the cylinder (A) is divided into N sections, so that each rod has a cross-sectional area of $(A/N)^2$ sq. km, so as the value of N in the program is increased the approximation of rods to line masses becomes more exact, and the jagged edge becomes smoother.

3. DATA INPUT

All input data are on punched cards.

First data card

This contains traverse and plotting details for the computer output. YS is the scale of the model to be computed; it is chosen so that the plotted profile has reasonable dimensions (Units km/in). XX1 defines the starting point of the traverse; this is the distance from the cylinder axis to the first observation (units km). MAXVAL is the absolute value of the minimum or maximum gravity value expected from the computations (units mgal). The vertical scale of the plotted output is then MAXVAL/5 (mgal/in). MAXVAL may be omitted, in which case, the scale is automatically set to a base of 10. DX (km) is the distance between observations along the traverse. DX may be omitted; this causes computations to be made at every half inch on the plotted profile or at YS/2 km intervals along the traverse. If DX is non-zero, but is set very small compared to XX1, computation time is large. In general XX1/Dx should be kept less than 30, and as small as possible. YT is the perpendicular distance from the traverse to the cylinder axis, so if the traverse passes over the cylinder centre, YT is taken as zero, or is left blank.

FIRST DATA CARD: YS, XX1, MAXVAL, DX, YT

100 Format : F3.0, F3.0, I3, F3.0, F3.0

YS and XX1 are essential data; the rest of the card may be left blank without causing the output to terminate.

Second data card

This defines the cylinder. A, D, and T are the cylinder radius, depth to top, and thickness respectively; all are expressed in km. P is the density contrast (+ or -) given in c.g.s., and S is an alphabetical identifier used to label the cylinder; it need not be present.

SECOND DATA CARD: A, D, T, P, S
A T P D S ← NEW

101 Format : F5.2, F5.2, F5.2, F5.2, A1

If more cylinders are required to construct the 3-D body, further data cards of the second type follow immediately, and are terminated by an EOF card. Up to 20 cylinders may be specified and used to construct a single 3-D body. The PL graph will consist of profiles which are due to each individual cylinder, plus a composite profile which is due to the additive effect of all the cylinders specified in the data.

Further models

If more than one model is to be computed in a single run, further data follow the EOF card. Both types of data must be specified for the new model, starting with traverse details (i.e. first data card) as described above. If no further models are to be computed, the EOF card is followed by a second EOF card, and then the EOD card.

Deck structure

See Appendix 4

Data sheet format

See Appendix 3

4. AN EXAMPLE - GOSSES BLUFF

The gravitational field around Gosses Bluff exhibits a certain amount of circular symmetry, so it is reasonable to construct a model composed of concentric cylinders, and to apply GRAVCYL to determine the accuracy of the model.

Plate 1 shows the observed gravity profiles over the Bluff to the north, and to the south. The model shown in this diagram is an attempt to explain the average observed profile; it is not a final geological interpretation, but it is chosen to illustrate the GRAVCYL method. The cylinder A represents an area of fractured rock, which gives rise to a density lower than the surrounding area. Cylinders B and C, when added, cancel in part and form an annulus of low density; to this, must be added the density of cylinder A. This annulus of relatively very low density corresponds to the walls of the Bluff.

Specifications of Model (Plate 1)

Cylinder	Radius	Depth	Thickness	Density
A	10.5	0.0	2.0	-0.06
B	2.4	0.0	1.5	-0.06
C	1.4	0.0	1.5	+0.06
D	4.5	5.0	3.0	+0.10

The starting point of the traverse (XX1) was 15.00, and the distance between observation points (DX) was 0.5. MAXVAL was set at 35, and YS at 5.

The time taken for computation of the gravity profile for the above model was 83 seconds, so that the time required for the computation of the gravity profile over a single cylinder (30 observation points) is about 20 seconds.

5. PROGRAM TESTING

Program GRAVCYL was tested by computing gravity values in a traverse over a cylinder, and comparing the results with those obtained by Nagy (1966). The cylinder had unit density, 100 metre radius, and 100 metre thickness. The depth to the cylinder was taken to be zero, and the distance between observations was 25 metres. The computation results appear in Table 1.

A mathematically exact value of gravity can be calculated at the axis of a cylinder, and for the above case, it is found to be 2.43106 mgal. It can be seen from Table 1 that Nagy's result is one percent too large, and the GRAVCYL result is 1.4 percent too small when N in the program is set at 25. When N is increased to 50, the GRAVCYL result at the axis becomes 2.42484 mgal. This represents an error of -0.25 percent. For most applications the 1.4 percent accuracy would be sufficient, and computing time is kept to a minimum.

6. DISCUSSION

Program GRAVCYL is presented here in basic form. It can be manipulated to give gravity profiles over a wide range of complex bodies, such as Gosses Bluff, and consequently is a useful functional program. However, the program could be further refined to sum cylinders whose centres are not on the traverse bisection and which are not concentric. This would further broaden the range of models for computation.

The program could be extended to provide for contouring of the gravity values calculated for the structure.

Further, by substituting the appropriate formulae it should be possible to calculate magnetic anomalies across cylindrical bodies.

7. REFERENCES

HAIGH, J.E., POLLARD, P.C., and WILLIAMS, J.P., (in prep.) - A computer program for the calculation of gravity and magnetic curves for two-dimensional bodies of arbitrary cross-section. Bur. Miner. Resour. Aust. Rec.

MORGAN, N.A., and GRANT, F.S., 1963 - High speed calculation of gravity and magnetic profiles across two dimensional bodies having an arbitrary cross-section. Geophysical Prospecting, 11, p. 10.

TALWANI, M., WORZEL, J.L., and LANDISMAN, M., 1959 - Rapid gravity computations for two dimensional bodies with application to the Mendocino submarine fracture zone. J. geophys. Res., 64, p. 49.

NAGY, D., 1966 - The evaluation of Heuman's lambda function and its application to calculate the gravitational effect of a right circular cylinder. Pure and Applied Geophys. 62, p. 5.

TABLE I

g_a = 2.43106 mgal (direct computation at axis)

DISTANCE FROM CYLINDER AXIS	NAGY (mgal)	N = 25 GRAVCYL (mgal)	DIFFERENCE (mgal)	PERCENTAGE DIFFERENCE	N = 50 GRAVCYL (mgal)
0.0 (m)	2.45496	2.42444	-0.03052	-1.26	2.42569
25	2.41164	2.37670	-0.03494	-1.47	2.37744
50	2.26854	2.23533	-0.03321	-1.49	2.24468
75	1.97299	1.96361	-0.00938	-0.48	1.95305
100	1.24811	1.35029	0.10218	7.56	1.29969
125	0.57363	0.60719	0.03356	5.52	0.58553
150	0.33152	0.34843	0.01691	4.85	0.33724
175	0.20817	0.21793	0.00976	4.48	0.21138
200	0.13867	0.14477	0.00610	4.21	0.14064
225	0.09676	0.10081	0.00405	4.02	0.09804
250	0.07010	0.07292	0.00282	3.88	0.07098
275	0.05238	0.05441	0.00203	3.74	0.05300
300	0.04015	0.04167	0.00152	3.65	0.04061
325	0.03145	0.03262	0.00117		0.03180
350	0.02509	0.02601	0.00092		0.02537
375	0.02034	0.02107	0.00073		0.02056
400	0.01672	0.01731	0.00059		0.01689
425	0.01391	0.01439	0.00048		0.01405
450	0.01170	0.01210	0.00040		0.01181
475	0.00993	0.01027	0.00034		0.01003
500	0.00850	0.00879	0.00029		0.00859

Radius (A) = 0.1 km

Thickness (T) = 0.1 km

Depth (D) = 0.0 km

Density (P) = 1 c.g.s.

APPENDIX 1: DERIVATION OF FORMULA



D: Depth to top of cylinder

T: Thickness of cylinder

R: Distance from observation point to mass element

m_1: Mass per unit length

g_v: Vertical component of gravity

g: Total gravity due to mass element m_1

G: Gravitational constant

$$g = Gm_1\left[\frac{1}{R^2}\right]$$

$$g_v = Gm_1\left[\frac{1}{R^2}\right]\cos\theta$$

$$= Gm_1\left[\frac{1}{R^2}\right]y/R$$

$$g_v = G\int_D^{(D+T)} \frac{m_1\, y\, dy}{(y^2 + x^2)^{3/2}}$$

$$= Gm_1\left[\frac{-1}{(y^2 + x^2)^{\frac{1}{2}}}\right]_D^{(D+T)}$$

$$= Gm_1\left(\frac{1}{R_1} - \frac{1}{R_2}\right)$$

Appendix II

```
      PROGRAM GRAVCYL
      DIMENSION VG(200),DIS(200),GO(20,200),WID(20)
      COMMON AX,AY,YS,XS,GO,DIS,VG,NL,NLA,KX,DX,WID
  105 FORMAT(I3,F3.0)
  104 FORMAT (30X,F8.6,2X,F8.6,5X,E20.5,E20.5)
  102 FORMAT(1X,F6.2,5X,E20.5)
  101 FORMAT (4F5.2,A1)
  100 FORMAT(2F3.0,I3,2F3.0)
C
   26 CONTINUE
      NLA=200
      KX=1
      READ 100,YS,XX1,MAXVAL,DX,YT
      IF(EOF,60)29,28
C     YS IS THE SCALE OF THE Y AXIS  (KM/IN)
C     XX1 DISTANCE OF FIRST OBSERVATION FROM CENTRE ON Y AXIS
C     MAXVAL IS ABSOLUTE VALUE OF MAX OR MIN GRAVITY VALUE EXPECTED (MGAL)
C     DX IS INCREMENT IN STEPS BETWEEN OBSERVATIONS
C     YT IS THE DISTANCE OF THE TRAVERSE FROM THE CYLINDER CENTRE
C
   28 CONTINUE
      GMAXV=0.0
      IF(DX.LT..0001)10,11
   10 DX=YS/2.0
   11 CONTINUE
      DO 31 JK=1,NLA
   31 VG(JK)=0.0
      KX=0
C
   25 READ 101,A,D,T,P,S$IF(EOF,60)23,24
C     A=RADIUS  D=DEPTH TO TOP  T=THICKNESS  (UNITS KM.)
C     P=DENSITY CONTRAST IN CGS .., S=ALPHA, IDENTIFIER
C
   24 N=25
      KX=KX+1
      NL=0
      X1=XX1 $ GV=0.0 $ G=0.667
      L=2*N+1 $ NK=N
         IF(YT.GT..001) NK=L
      RN=A/N
      P=P*RN*RN
      D=D+A/1000
         KSHIFT=(YT+RN/2.0)/RN
         YT=KSHIFT*RN
      ARN=A+RN/2.0
      ALE=ARN*ARN
      AN=A+RN  $  YTR=YT+RN/2
         YTRA=YTR+A
         D2D=D*D  $  D2T=(D+T)*(D+T)
      KAT=(XX1/DX)*2.0+2.0
      X1=X1+DX
   22 X1=X1-DX $ GV=0.00
      NL=NL+1
      NOL=KAT-NL
      DIS(NL)=X1 $ DIS(NOL)=-DIS(NL)
C     --------------------------------------------------------------
      DO 20 K1=1,L
      DO 20 K2=1,NK
      GXY=0.00 $ R1=0.00 $ R2=0.00
      X=K1*RN-AN
         Y=YTRA-K2*RN
      Y2=Y*Y
      YYT=Y-YT  $  Y2T=YYT*YYT
      IF((X*X+Y2T).GT.ALE)20,21
   21 X1X=X1-X
      X1X2=X1X*X1X
      XY=Y2+X1X2
         R1=SQRT(XY+D2D)
         R2=SQRT(XY+D2T)
      GXY=G*P*(R2-R1)/(R1*R2)
   19 GV=GV+GXY*1E+1
   20 CONTINUE
C     --------------------------------------------------------------
         IF(YT.LT..001) GV=2.0*GV
      WID(KX)=S
      GO(KX,NOL)=GO(KX,NL)=GV
      VG(NL)=VG(NL)+GV
      VG(NOL)=VG(NL)
      GN=ABS(VG(NL))
      IF(GN.GT.GMAXV)12,13
   12 GMAXV=GN
   13 CONTINUE
      IF(X1.LE.0)25,22
   23 IF(MAXVAL.LT..0001)14,15
C     --------------------------------------------------------------
   14 AI=ALOG10(GMAXV)
      IA=AI
      IF(AI.LT.0.) IA=IA-1
      POWER=10.0**IA
      NA=(GMAXV/POWER + 0.5)
      IF(NA/3 - 1)60,61,62
   60  XS=0.5*POWER $ GO TO 63
   61  XS=POWER $ GO TO 63
   62  XS=2.0*POWER $ GO TO 63
   15 XS=(MAXVAL-1)/5+1
C     --------------------------------------------------------------
   63 AY=-XX1
      AX=-5.0*XS
C      AY,AX  DEFINE ORIGIN ON P@ PLOTTER, TAKE ORIGIN AT CENTRE OF P@
      CALL PLOTOUT
      CALL PRINTOUT
      GO TO 26
   29 CONTINUE
      END
```

Appendix II Cont

```
      SUBROUTINE PLOTOUT
      DIMENSION VG(200),DIS(200),GO(20,200),WID(20)
      COMMON AX,AY,YS,XS,GO,DIS,VG,NL,NLA,KX,DX,WID
C     ------------------------------------------------------------
C       DRAWS AXES
      JAX=(-AX/XS)*2
      JAY=(-AY/YS)*2
      BL=YS/10.0
      CALL PLOT (XS,YS,2)
      CALL PLOT (AX,AY,1)
      CALL PLOT (0.,AY,3)
      DO 97 IAB=1,JAY
      ZK=AY+IAB*YS
      CALL PLOT (0.,ZK,4)
      CALL PLOT (BL,ZK,4)
      ENCODE (5,98,BUF)ZK
   98 FORMAT (F5.1)
      CALL TEXT (BUF,5,1)
   97 CALL PLOT (0.,ZK,3)
      CALL PLOT (0.,0.,3)
      DO 96 IBB=1,9
      ZOTY=(IBB-5)*XS
      CALL PLOT (ZOTY,0.,4)
      CALL PLOT (ZOTY,BL,4)
      CALL PLOT ((ZOTY-XS*0.2),0.0,3)
      IF(XS.LT.1.0)51,52
   51 ENCODE (5,94,BUF)ZOTY $ CALL TEXT (BUF,5,2) $ GO TO 96
   94 FORMAT(F5.2)
   52 ENCODE (4,95,BUF)ZOTY
   95 FORMAT (F4.0)
      CALL TEXT (BUF,4,2)
   96 CALL PLOT (ZOTY,0.,3)
C     ------------------------------------------------------------
  103 FORMAT (A1)
      MIP=YS/(2*DX)+1
      TST=MIP*DX
   30 NLA=2*NL-1
C       PLOT INDIVIDUAL PROFILES
      DO 4 J=1,KX
      S=WID(J)
      FL=0
      DO 4 I=1,NLA
      GV=GO(J,I) $ X1=DIS(I)
      IF (GV.GT.-AX.OR.GV.LT.AX)4,1
    1 IF(FL.EQ.0)2,3
    2 CALL PLOT (GV,X1,3) $ GO TO 4
    3 CALL PLOT (GV,X1,4)
      IF(X1.EQ.TST)7,8
    7 ENCODE (1,103,BUF)S
      CALL TEXT (BUF,1,2)
      CALL PLOT (GV,X1,3)
    8 CONTINUE
    4 FL=1
C     ------------------------------------------------------------
C       PLOT COMPOSITE PROFILE
      CALL PLOT (VG(1),XX1,3)
      DO 27 I=1,NLA
      IF (VG(I).GT.-AX.OR.VG(I).LT.AX)27,6
    6 CALL PLOT(VG(I),DIS(I),4)
   27 CONTINUE
      CALL PLOTCHOP
      RETURN
      END
```

```
      SUBROUTINE PRINTOUT
      DIMENSION VG(200),DIS(200),GO(20,200),WID(20)
      COMMON AX,AY,YS,XS,GO,DIS,VG,NL,NLA,KX,DX,WID
   99 FORMAT (5X,E14.7,5X,E14.7)
      DO 33 I=1,NLA
      PRINT 99,DIS(I),VG(I)
   33 CONTINUE
      RETURN
      END
```

PLATE 1



GOSSES BLUFF MODEL AND COMPUTED GRAVITY PROFILES

To accompany Record No. 1971/74

G65-311A

PLATE 2



PROGRAM FLOW CHART

To accompany Record No 1971/74 G65-312A

PLATE 3

CSIRO DATA SHEET

PROGRAM	NAME	JOB	DATE	PAGE NO.

COMMENT

YS XX1 MAXVAL DX YT

F3.0 F3.0 I3 F3.0

005 015 035 0·5

NOTE: MAXVAL AND DX MAY BE LEFT BLANK ALSO

F5.2

10·50	00·00	02·00	−0·06	A
02·40	00·00	01·50	−0·06	B
01·40	00·00	01·50	+0·06	C
04·50	05·00	03·00	+0·10	D

RADIUS DEPTH THICKNESS DENSITY IDENTIFIER

PLATE 3. DATA PREPARATION FOR MODEL SHOWN IN PLATE 1

PLATE 4



DECK STRUCTURE

To accompany Record No. 1971/74 G65-313A